Lesson 27: Asymptotic Series

```
> restart;
```

Asymptotic series for an exponential integral

We were looking at the following integral as $x \to \infty$.

```
> J := Int(exp(-t)/t, t=x..infinity);
  Jv:= value(J);
```

$$J := \int_x^\infty \frac{e^{-t}}{t}\, dt$$

$$Jv := \mathrm{Ei}(1, x)$$

```
> asympt(%,x,10);
```

$$\frac{e^{-x}}{x} - \frac{e^{-x}}{x^2} + \frac{2\, e^{-x}}{x^3} - \frac{6\, e^{-x}}{x^4} + \frac{24\, e^{-x}}{x^5} - \frac{120\, e^{-x}}{x^6} + \frac{720\, e^{-x}}{x^7} - \frac{5040\, e^{-x}}{x^8} + \frac{40320\, e^{-x}}{x^9} + \mathrm{O}\left(\frac{1}{x^{10}}\right)$$

Our series was $e^x \left(\int_x^\infty \frac{e^{-t}}{t}\, dt \right) = \sum_{k=0}^{\infty} \frac{(-1)^k\, k!}{x^{k+1}}$

which is an asymptotic series that diverges for all x.

Let's see how well the partial sums of our asymptotic series do at approximating the original integral J.

```
> for count from 1 to 30 do
    PS[count]:= exp(-x)*add((-1)^k*k!/x^(1+k),k=0..count-1)
  end do:
```

The more terms we have in the series, the better it is at approximating J when x is large. But how large is "large" depends on n. For any particular x, the approximations typically get better for a while, but then eventually get worse, because the series diverges. Here are the errors for $x = 5$, for example.

```
> Errors5:=[seq([n, evalf(eval(Jv-PS[n],x=5))], n=1..30)];
```

Errors5 := [[1, -0.000199293809], [2, 0.000070224071], [3, -0.000037583081], [4, 0.000027101210], [5, -0.000024646223], [6, 0.000027101210], [7, -0.000034995709], [8, 0.000051939978], [9, -0.000087157122], [10, 0.0001632176580], [11, -0.000337531901], [12, 0.0007641171290], [13, -0.001879840543], [14, 0.004994449405], [15, -0.01425356245], [16, 0.04349047312], [17, -0.1412904407], [18, 0.4869646663], [19, -1.774753718], [20, 6.819776146], [21, -27.55834330], [22, 116.8297584], [23, -518.4778891], [24, 2403.937289], [25, -11623.65557], [26, 58514.30872], [27, $-3.062031056\ 10^5$], **(1.1)**

$[28, 1.663270931\ 10^6], [29, -9.365783676\ 10^6], [30, 5.460273305\ 10^7]]$

What's the smallest error (in absolute value)? First get the absolute values.

```
> map(t -> abs(t[2]), Errors5);
```

[0.000199293809, 0.000070224071, 0.000037583081, 0.000027101210, 0.000024646223, 0.000027101210, 0.000034995709, 0.000051939978, 0.000087157122, 0.0001632176580, 0.000337531901, 0.0007641171290, 0.001879840543, 0.004994449405, 0.01425356245, 0.04349047312, 0.1412904407, 0.4869646663, 1.774753718, 6.819776146, 27.55834330, 116.8297584, 518.4778891, 2403.937289, 11623.65557, 58514.30872, $3.062031056\ 10^5$, $1.663270931\ 10^6$, $9.365783676\ 10^6$, $5.460273305\ 10^7$] **(1.2)**

Then take the minimum using **min**.

```
> min(%);
```

0.000024646223 **(1.3)**

Which entry had this (or -this)?

```
> select(has,Errors5,{%,-%});
```

[[5, -0.000024646223]] **(1.4)**

Let's try some animations. In the n'th frame, I'll plot J - PS[n] from x = 1 to 10.

```
> with(plots):
display([seq(plot(Jv - PS[n], x=1..10,title=('n'=n)), n=1..30)], insequence=true,view=[1..10,-1..1]);
```



That's maybe a pessimistic view: it shows really bad approximations for larger and larger x as n increases.
It doesn't show the good approximations very well.
A better idea, showing both the good and bad, comes from a logarithmic plot. The **logplot** command in the **plots** package plots an expression (which should be positive) using a logarithmic scale for the y axis. Here the approximation is better if the value of the log is more negative.

```
> display([seq(logplot(abs(Jv - PS[n]), x=1..20, title=('n'=n)
  ), n=1..30)],insequence=true,view=[1..20,10^(-19)..1]);
```



Approximating solutions of an equation

Here is an example that actually came up a while ago in a question I was asked by a friend. The problem arose from research having to do with approximation of functions, and turns out to require solving the equation $\cos(x)\cosh(x) = -1$. Note: cosh is the hyperbolic cosine function

```
> cosh(x) = convert(cosh(x),exp);
```

$$\cosh(x) = \frac{1}{2}\,e^{x} + \frac{1}{2}\,e^{-x} \tag{2.1}$$

```
> eq:= cos(x)*rhs(%)=-1;
```

$$eq := \cos(x)\left(\frac{1}{2}\,e^{x} + \frac{1}{2}\,e^{-x}\right) = -1 \tag{2.2}$$

```
> solve(eq, x);
```

$$\mathit{RootOf}\left(\cos(\_Z)\left(e^{\_Z}\right)^{2} + \cos(\_Z) + 2\,e^{\_Z}\right) \tag{2.3}$$

It seems the equation can't be solved in closed form. But it's easy to see that there will be a solution in each interval $n\,\pi < x < (n+1)\,\pi$ for integers n, because the left side will be positive at

$x = n\,\pi$ for even n and less than -1 for odd n.

```
> plot([cos(x)*cosh(x),-1],x=0..3*Pi,-5..5);
```



In fact there is exactly one solution in each of these intervals: call it x_n. I want to find good approximations for these solutions. Of course **fsolve** can do it for any particular n, but I want a general formula.

We should expect x_n to be near $\left(n+\frac{1}{2}\right)\pi$, where the left side of the equation is 0. So I'll write $x = \left(n+\frac{1}{2}\right)\pi + t$.

```
> eq1:= expand(eval(eq,x=(n+1/2)*Pi+t));
```

$$eq1 := -\frac{1}{2}\cos(\pi\,n)\sin(t)\,e^{\pi\,n}\sqrt{e^{\pi}}\,e^{t} - \frac{1}{2}\,\frac{\cos(\pi\,n)\sin(t)}{e^{\pi\,n}\sqrt{e^{\pi}}\,e^{t}} - \frac{1}{2}\sin(\pi\,n)\cos(t)\,e^{\pi\,n}\sqrt{e^{\pi}}\,e^{t} - \frac{1}{2}\,\frac{\sin(\pi\,n)\cos(t)}{e^{\pi\,n}\sqrt{e^{\pi}}\,e^{t}} = -1 \tag{2.4}$$

Doesn't Maple know that $\sin(\pi\,n) = 0$? It does if we tell it that n is an integer.

```
> eq1:= eq1 assuming n::integer;
```

$$eq1 := -\frac{1}{2}\,(-1)^n \sin(t)\, e^{\pi n} \sqrt{e^{\pi}}\, e^t - \frac{1}{2}\,\frac{(-1)^n \sin(t)}{e^{\pi n}\sqrt{e^{\pi}}\, e^t} = -1 \qquad \textbf{(2.5)}$$

Let's make a new variable $u = e^{\pi n}$.

```
> eq2:= eval(eq1,exp(Pi*n)=u);
```

$$eq2 := -\frac{1}{2}\,(-1)^n \sin(t)\, u \sqrt{e^{\pi}}\, e^t - \frac{1}{2}\,\frac{(-1)^n \sin(t)}{u\sqrt{e^{\pi}}\, e^t} = -1 \qquad \textbf{(2.6)}$$

How does t behave when u is large?

```
> asympt(RootOf(eq2,t),u,15);
```

$$\frac{2\,e^{-\frac{1}{2}\pi}(-1)^{-n}}{u} - \frac{4\,e^{-\pi}(-1)^{-2n}}{u^2} + \frac{2}{3}\,\frac{e^{-\frac{3}{2}\pi}\left(20\,(-1)^{-3n} - 3\,(-1)^{-n}\right)}{u^3} \qquad \textbf{(2.7)}$$
$$+ \frac{16}{3}\,\frac{e^{-2\pi}\left(-10\,(-1)^{-4n} + 3\,(-1)^{-2n}\right)}{u^4}$$
$$- \frac{2}{15}\,\frac{e^{-\frac{5}{2}\pi}\left(-1768\,(-1)^{-5n} + 780\,(-1)^{-3n} - 15\,(-1)^{-n}\right)}{u^5}$$
$$- \frac{4}{3}\,\frac{e^{-3\pi}\left(832\,(-1)^{-6n} - 480\,(-1)^{-4n} + 27\,(-1)^{-2n}\right)}{u^6}$$
$$+ \frac{2}{63}\,\frac{1}{u^7}\Big(e^{-\frac{7}{2}\pi}\big(-63\,(-1)^{-n} + 171680\,(-1)^{-7n} - 121800\,(-1)^{-5n}$$
$$+ 12600\,(-1)^{-3n}\big)\Big)$$
$$+ \frac{64}{63}\,\frac{e^{-4\pi}\left(-3570\,(-1)^{-4n} + 63\,(-1)^{-2n} - 27200\,(-1)^{-8n} + 22848\,(-1)^{-6n}\right)}{u^8}$$
$$- \frac{2}{63}\,\frac{1}{u^9}\Big(e^{-\frac{9}{2}\pi}\big(4380768\,(-1)^{-7n} - 929880\,(-1)^{-5n} + 34440\,(-1)^{-3n}$$
$$- 4519840\,(-1)^{-9n} - 63\,(-1)^{-n}\big)\Big) - \frac{4}{2835}\,\frac{1}{u^{10}}\Big(e^{-5\pi}\big(538150912\,(-1)^{-10n}$$
$$- 590653440\,(-1)^{-8n} + 159606720\,(-1)^{-6n} - 9828000\,(-1)^{-4n} + 70875\,(-1)^{-2n}\big)$$
$$\Big) + \frac{2}{6237}\,\frac{1}{u^{11}}\Big(e^{-\frac{11}{2}\pi}\big(-461621160\,(-1)^{-5n} + 12722612864\,(-1)^{-11n}$$
$$- 15588349920\,(-1)^{-9n} + 5134985856\,(-1)^{-7n} + 7609140\,(-1)^{-3n} - 6237\,($$
$$-1)^{-n}\big)\Big) + \frac{16}{693}\,\frac{1}{u^{12}}\Big(e^{-6\pi}\big(-506373120\,(-1)^{-8n} + 61264896\,(-1)^{-6n}$$
$$- 961445888\,(-1)^{-12n} + 1300316160\,(-1)^{-10n} - 1794870\,(-1)^{-4n} + 6237\,($$

$$-1)^{-2n})\Big) - \frac{2}{243243}\frac{1}{u^{13}}\Big(e^{-\frac{13}{2}\pi}\big(21960611587200\,(-1)^{-11n} - 9882275214240\,(-1)^{-9n} + 1523139666048\,(-1)^{-7n} - 14845168672000\,(-1)^{-13n} - 68698309680\,(-1)^{-5n} + 578918340\,(-1)^{-3n} - 243243\,(-1)^{-n}\big)\Big) - \frac{4}{18711}\frac{1}{u^{14}}\Big(e^{-7\pi}\big(3167921766400\,(-1)^{-14n} - 5087309660160\,(-1)^{-12n} + 2599512115200\,(-1)^{-10n} - 490816972800\,(-1)^{-8n} + 31099178880\,(-1)^{-6n} - 488980800\,(-1)^{-4n} + 916839\,(-1)^{-2n}\big)\Big) + O\left(\frac{1}{u^{15}}\right)$$

```
> S:= map(simplify,%) assuming n::integer;
```

$$S := \frac{2\,e^{-\frac{1}{2}\pi}\,(-1)^n}{u} - \frac{4\,e^{-\pi}}{u^2} + \frac{34}{3}\frac{e^{-\frac{3}{2}\pi}\,(-1)^n}{u^3} - \frac{112}{3}\frac{e^{-2\pi}}{u^4} + \frac{2006}{15}\frac{e^{-\frac{5}{2}\pi}\,(-1)^n}{u^5} - \frac{1516}{3}\frac{e^{-3\pi}}{u^6} + \frac{124834}{63}\frac{e^{-\frac{7}{2}\pi}\,(-1)^n}{u^7} - \frac{502976}{63}\frac{e^{-4\pi}}{u^8} + \frac{2069150}{63}\frac{e^{-\frac{9}{2}\pi}\,(-1)^n}{u^9} - \frac{389388268}{2835}\frac{e^{-5\pi}}{u^{10}} + \frac{518637298}{891}\frac{e^{-\frac{11}{2}\pi}\,(-1)^n}{u^{11}} - \frac{1728425360}{693}\frac{e^{-6\pi}}{u^{12}} + \frac{2623624535150}{243243}\frac{e^{-\frac{13}{2}\pi}\,(-1)^n}{u^{13}} - \frac{879673454236}{18711}\frac{e^{-7\pi}}{u^{14}} + O\left(\frac{1}{u^{15}}\right) \quad (2.8)$$

```
> evalf(%);
```

$$\frac{0.4157591526\,(-1.)^n}{u} - \frac{0.1728556730}{u^2} + \frac{0.1018106315\,(-1.)^n}{u^3} - \frac{0.06971786191}{u^4} + \frac{0.05191570840\,(-1.)^n}{u^5} - \frac{0.04078015616}{u^6} + \frac{0.03324107785\,(-1.)^n}{u^7} - \frac{0.02784205555}{u^8} + \frac{0.02380991440\,(-1.)^n}{u^9} - \frac{0.02069893635}{u^{10}} + \frac{0.01823543356\,(-1.)^n}{u^{11}} - \frac{0.01624273926}{u^{12}} + \frac{0.01460209037\,(-1.)^n}{u^{13}} - \frac{0.01323091894}{u^{14}} + O\left(\frac{1}{u^{15}}\right) \quad (2.9)$$

It looks like this should converge for approximately $u > 1$, thus maybe not for $n = 0$ but certainly for $n \geq 1$.

```
> Sn := eval(S, u = exp(Pi*n));
```

$$Sn := \frac{2\,e^{-\frac{1}{2}\pi}\,(-1)^n}{e^{\pi n}} - \frac{4\,e^{-\pi}}{(e^{\pi n})^2} + \frac{34}{3}\,\frac{e^{-\frac{3}{2}\pi}\,(-1)^n}{(e^{\pi n})^3} - \frac{112}{3}\,\frac{e^{-2\pi}}{(e^{\pi n})^4} + \frac{2006}{15}\,\frac{e^{-\frac{5}{2}\pi}\,(-1)^n}{(e^{\pi n})^5} - \frac{1516}{3}\,\frac{e^{-3\pi}}{(e^{\pi n})^6} + \frac{124834}{63}\,\frac{e^{-\frac{7}{2}\pi}\,(-1)^n}{(e^{\pi n})^7} - \frac{502976}{63}\,\frac{e^{-4\pi}}{(e^{\pi n})^8} + \frac{2069150}{63}\,\frac{e^{-\frac{9}{2}\pi}\,(-1)^n}{(e^{\pi n})^9} - \frac{389388268}{2835}\,\frac{e^{-5\pi}}{(e^{\pi n})^{10}} + \frac{518637298}{891}\,\frac{e^{-\frac{11}{2}\pi}\,(-1)^n}{(e^{\pi n})^{11}} - \frac{1728425360}{693}\,\frac{e^{-6\pi}}{(e^{\pi n})^{12}} + \frac{2623624535150}{243243}\,\frac{e^{-\frac{13}{2}\pi}\,(-1)^n}{(e^{\pi n})^{13}} - \frac{879673454236}{18711}\,\frac{e^{-7\pi}}{(e^{\pi n})^{14}} + O\left(\frac{1}{(e^{\pi n})^{15}}\right) \tag{2.10}$$

Even for $n = 1$, the first four terms are already quite accurate.

```
> S4:= convert(asympt(S,u,5),polynom);
```

$$S4 := \frac{2\,e^{-\frac{1}{2}\pi}\,(-1)^n}{u} - \frac{4\,e^{-\pi}}{u^2} + \frac{34}{3}\,\frac{e^{-\frac{3}{2}\pi}\,(-1)^n}{u^3} - \frac{112}{3}\,\frac{e^{-2\pi}}{u^4} \tag{2.11}$$

```
> Digits:= 15; x1:= fsolve(eq, x=Pi .. 2*Pi);
```

$$Digits := 15$$

$$x1 := 4.69409113297417 \tag{2.12}$$

```
> x1app:= evalf(3/2*Pi+eval(S4,{u=exp(Pi),n=1}));
```

$$x1app := 4.69409114107324 \tag{2.13}$$

```
> x1-x1app;
```

$$-8.09907\ 10^{-9} \tag{2.14}$$

The Euler-Maclaurin summation formula

```
> restart;
```

Suppose we have some nice function $f(x)$, and we want $F(x)$ so that $F(x+1) - F(x) = f(x)$. This is sometimes called an **anti-difference** of $f(x)$ (by analogy with anti-derivative). For example, if we're just concerned with integers, we could take $\sum_{k=0}^{x-1} f(k)$. Our main application is going to be in looking at sums like that. Note that, just like anti-derivatives, anti-differences are not unique, because you can always add a constant. In fact, you can add any function that's periodic with period 1.

On a purely formal level, we can proceed as follows. Using Taylor series,

$$F(x+1) = \sum_{n=0}^{\infty} \frac{D^{(n)}(F)(x)}{n!} = e^{D}(F)(x)$$

(for the $n = 0$ term, $D^{(0)}$ is the identity operator, which we'll write as 1).

So we can say we want to solve $\left(e^{D} - 1\right)(F)(x) = f(x)$.

What this really means is: expand $e^{d} - 1$ as a Maclaurin series in d, and replace d^n by $D^{(n)}$.

Well, we might write $F(x) = \left(e^{D} - 1\right)^{-1}(f)(x)$. Now $\left(e^{d} - 1\right)^{-1}$ has a series in powers of d:

```
> series(1/(exp(d)-1),d,10);
```

$$d^{-1} - \frac{1}{2} + \frac{1}{12}\,d - \frac{1}{720}\,d^3 + \frac{1}{30240}\,d^5 - \frac{1}{1209600}\,d^7 + O\left(d^8\right) \tag{3.1}$$

The differentiation operator D doesn't really have an inverse, but we can say integration is the closest thing there is to one. So we might try something like this as an approximation to an antidifference of $f(x)$ (neglecting terms involving the 8th and higher derivatives):

```
> G:= int(f(x),x) - 1/2*f(x) + 1/12*D(f)(x) - 1/720*(D@@3)(f)
  (x) + 1/30240*(D@@5)(f)(x) - 1/1209600*(D@@7)(f)(x);
```

$$G := \int f(x)\,dx - \frac{1}{2}\,f(x) + \frac{1}{12}\,D(f)(x) - \frac{1}{720}\,D^{(3)}(f)(x) + \frac{1}{30240}\,D^{(5)}(f)(x) - \frac{1}{1209600}\,D^{(7)}(f)(x) \tag{3.2}$$

Amazingly enough, this works. For example, try it on a polynomial of degree 7 (where those higher derivatives are 0) and it should work perfectly.

```
> f7:= unapply(add(a[j]*x^j,j=0..7),x);
```

$$f7 := x \rightarrow a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6 + a_7 x^7 \tag{3.3}$$

```
> G7:= unapply(eval(G,f=f7),x);
```

$$G7 := x \rightarrow a_0 x + \frac{1}{2}\,a_1 x^2 + \frac{1}{3}\,a_2 x^3 + \frac{1}{4}\,a_3 x^4 + \frac{1}{5}\,a_4 x^5 + \frac{1}{6}\,a_5 x^6 + \frac{1}{7}\,a_6 x^7 + \frac{1}{8}\,a_7 x^8 + \frac{1}{6}\,a_2 x + \frac{1}{4}\,a_3 x^2 + \frac{1}{3}\,a_4 x^3 + \frac{5}{12}\,a_5 x^4 + \frac{1}{2}\,a_6 x^5 + \frac{7}{12}\,a_7 x^6 - \frac{1}{30}\,a_4 x - \frac{1}{12}\,a_5 x^2 - \frac{1}{6}\,a_6 x^3 - \frac{7}{24}\,a_7 x^4 + \frac{1}{42}\,a_6 x + \frac{1}{12}\,a_7 x^2 - \frac{1}{2}\,a_1 x - \frac{1}{2}\,a_2 x^2 - \frac{1}{2}\,a_3 x^3 - \frac{1}{2}\,a_4 x^4 - \frac{1}{2}\,a_5 x^5 - \frac{1}{2}\,a_6 x^6 - \frac{1}{2}\,a_7 x^7 - \frac{1}{2}\,a_0 + \frac{1}{12}\,a_1 - \frac{1}{120}\,a_3 + \frac{1}{252}\,a_5 - \frac{1}{240}\,a_7 \tag{3.4}$$

```
> normal(G7(t+1)-G7(t)) = f7(t);
```

$$a_1 t + a_2 t^2 + a_3 t^3 + a_4 t^4 + a_5 t^5 + a_6 t^6 + a_7 t^7 + a_0 = a_1 t + a_2 t^2 + a_3 t^3 + a_4 t^4 + a_5 t^5 + a_6 t^6 + a_7 t^7 + a_0 \tag{3.5}$$

The coefficients in the Maclaurin series for $\frac{d}{e^{d} - 1}$ turn out to be expressed in terms of Bernoulli numbers:

```
> FunctionAdvisor(identities,bernoulli(n))[1,2];
```

$$\frac{\_t1}{e^{\_t1}-1} = \sum_{n=0}^{\infty} \frac{\text{bernoulli}(n)\ \_t1^n}{n!} \tag{3.6}$$

so our formula can be written as

$$F(x) = \int f(x)\,dx + C + \sum_{n=1}^{\infty} \frac{\text{bernoulli}(n)\ D^{(n-1)}(f)(x)}{n!}$$

This is usually a divergent series, so what we really have is (for fixed N)

$$F(x) = \int f(x)\,dx + C + \sum_{n=1}^{N} \frac{\text{bernoulli}(n)\ D^{(n-1)}(f)(x)}{n!} + R_N(x)$$

where the remainder $R_N(x)$ depends on $D^{(N)}(f)$: if $|D^{(N)}(f)(t)| \le K(x)$ for $x \le t \le x+1$, then $R_N(x) = O(K(x))$ as $x \to \infty$.

Here are the first few Bernoulli numbers:

```
> seq(b(n)=bernoulli(n),n=0..20);
```

$$b(0)=1, b(1)=-\frac{1}{2}, b(2)=\frac{1}{6}, b(3)=0, b(4)=-\frac{1}{30}, b(5)=0, b(6)=\frac{1}{42}, b(7)=0, b(8)=-\frac{1}{30}, b(9)=0, b(10)=\frac{5}{66}, b(11)=0, b(12)=-\frac{691}{2730}, b(13)=0, b(14)=\frac{7}{6}, b(15)=0, b(16)=-\frac{3617}{510}, b(17)=0, b(18)=\frac{43867}{798}, b(19)=0, b(20)=-\frac{174611}{330} \tag{3.7}$$

Except for n=1, the odd ones are 0. That's because $\frac{d}{e^d-1} + \frac{d}{2}$ is an even function.

```
> q:= d -> d/(exp(d)-1)+d/2;
```

$$q := d \to \frac{d}{e^d-1} + \frac{1}{2}d \tag{3.8}$$

```
> simplify(q(d)-q(-d));
```

$$0 \tag{3.9}$$

Maple has a command for the Euler-Maclaurin formula: **eulermac**.

```
> eulermac(f(x), x, 7);
```

$$\int f(x)\,dx - \frac{1}{2}f(x) + \frac{1}{12}\frac{d}{dx}f(x) - \frac{1}{720}\frac{d^3}{dx^3}f(x) + \frac{1}{30240}\frac{d^5}{dx^5}f(x) - \frac{1}{1209600}\frac{d^7}{dx^7}f(x) + O\left(\frac{d^9}{dx^9}f(x)\right)$$

Now, what can we do with the Euler-Maclaurin formula?

Sum of a slowly convergent series

The series

$\sum_{k=1}^{\infty} \frac{1}{k^2+\ln(k)}$ converges, but slowly. In Lesson 24 we developed ways of getting estimates for such series using upper and lower bounds. The best we had was to estimate the tail of the series (when f'' is decreasing) as

$$\int_{N+\frac{1}{2}}^{\infty} f(t)\,\mathrm{d}t + \frac{f'\left(N-\frac{1}{2}\right)}{24} < \sum_{n=N+1}^{\infty} f(n) < \int_{N+\frac{1}{2}}^{\infty} f(t)\,\mathrm{d}t + \frac{f'\left(N+\frac{3}{2}\right)}{24}$$

With the Euler-Maclaurin formula, we can do better. Note that $T(N) = \sum_{n=N}^{\infty} f(n)$ is an antidifference of $-f(N)$, since $T(N+1) - T(N) = -f(N)$. So according to Euler-Maclaurin:

```
> T(x) = eulermac(-f(x),x);
```

$$T(x) = \int (-f(x))\,\mathrm{d}x + \frac{1}{2} f(x) - \frac{1}{12} \frac{\mathrm{d}}{\mathrm{d}x} f(x) + \frac{1}{720} \frac{\mathrm{d}^3}{\mathrm{d}x^3} f(x) - \frac{1}{30240} \frac{\mathrm{d}^5}{\mathrm{d}x^5} f(x) + \mathrm{O}\left(\frac{\mathrm{d}^7}{\mathrm{d}x^7} f(x)\right)$$

We're interested in cases where $f(x)$ and its derivatives go to 0 as $x \to \infty$, and so of course does $T(x)$. So the antiderivative in this formula should be the one that goes to 0 as $x \to \infty$, namely $\int_{x}^{\infty} f(t)\,\mathrm{d}t$.

Thus we get an asymptotic formula for $T(x)$:

$$T(x) = \int_{x}^{\infty} f(t)\,\mathrm{d}t - \left(\sum_{n=1}^{N} \frac{\mathrm{bernoulli}(n)\, \mathrm{D}^{(n-1)}(f)(x)}{n!}\right) + \mathit{Remainder}.$$

I'll try this first with $f(x) = \frac{1}{x(x+1)}$ (which is a function for which both the sum and the integral can be done in closed form):

```
> f1:= x -> 1/(x*(x+1));
  Sum(f1(j),j=x..infinity) = sum(f1(j),j=x..infinity);
  Int(f1(t),t=x..infinity) = int(f1(t),t=x..infinity) assuming
  x > 0;
```

$$f1 := x \to \frac{1}{x(x+1)}$$

$$\sum_{j=x}^{\infty} \frac{1}{j(j+1)} = \frac{1}{x}$$

$$\int_{x}^{\infty} \frac{1}{t(t+1)}\,\mathrm{d}t = \ln(1+x) - \ln(x)$$

```
> Remainder:= sum(f1(j),j=x..infinity) - int(f1(t),t=x..
  infinity) + add(bernoulli(n)*(D@@(n-1))(f1)(x)/n!,n=1..10)
```

```
assuming x > 0;
```

$$Remainder := -\frac{1}{2x(1+x)} - \frac{1}{12x^2(1+x)} - \frac{1}{12x(1+x)^2} + \frac{1}{120x^4(1+x)} + \frac{1}{120x^3(1+x)^2} + \frac{1}{120x^2(1+x)^3} + \frac{1}{120x(1+x)^4} - \frac{1}{252x^6(1+x)} - \frac{1}{252x^5(1+x)^2} - \frac{1}{252x^4(1+x)^3} - \frac{1}{252x^3(1+x)^4} - \frac{1}{252x^2(1+x)^5} - \frac{1}{252x(1+x)^6} + \frac{1}{240x^8(1+x)} + \frac{1}{240x^7(1+x)^2} + \frac{1}{240x^6(1+x)^3} + \frac{1}{240x^5(1+x)^4} + \frac{1}{240x^4(1+x)^5} + \frac{1}{240x^3(1+x)^6} + \frac{1}{240x^2(1+x)^7} + \frac{1}{240x(1+x)^8} - \frac{1}{132x^9(1+x)^2} - \frac{1}{132x^8(1+x)^3} - \frac{1}{132x^7(1+x)^4} - \frac{1}{132x^6(1+x)^5} - \frac{1}{132x^5(1+x)^6} - \frac{1}{132x^4(1+x)^7} - \frac{1}{132x^3(1+x)^8} - \frac{1}{132x^2(1+x)^9} - \frac{1}{132x(1+x)^{10}} - \frac{1}{132x^{10}(1+x)} + \frac{1}{x} - \ln(1+x) + \ln(x) \tag{4.1}$$

```
> asympt(%,x,15);
```

$$-\frac{691}{2730x^{13}} + \frac{691}{420x^{14}} + O\left(\frac{1}{x^{15}}\right) \tag{4.2}$$

There's also a form of eulermac that let's you specify that you want a formula for $\sum_{j=a}^{b} f(j)$: you just say eulermac(f(j),j=a..b). Thus:

```
> eulermac(f1(t),t=x..infinity,10);
```

$$\frac{1}{2x(1+x)} + \frac{1}{12x^2(1+x)} + \frac{1}{12x(1+x)^2} - \frac{1}{120x^4(1+x)} - \frac{1}{120x^3(1+x)^2} - \frac{1}{120x^2(1+x)^3} - \frac{1}{120x(1+x)^4} + \frac{1}{252x^6(1+x)} + \frac{1}{252x^5(1+x)^2} + \frac{1}{252x^4(1+x)^3} + \frac{1}{252x^3(1+x)^4} + \frac{1}{252x^2(1+x)^5} + \frac{1}{252x(1+x)^6} - \frac{1}{240x^8(1+x)} - \frac{1}{240x^7(1+x)^2} - \frac{1}{240x^6(1+x)^3} - \frac{1}{240x^5(1+x)^4} - \frac{1}{240x^4(1+x)^5} - \frac{1}{240x^3(1+x)^6} - \frac{1}{240x^2(1+x)^7} - \frac{1}{240x(1+x)^8} + \frac{1}{132x^9(1+x)^2} + \frac{1}{132x^8(1+x)^3} + \frac{1}{132x^7(1+x)^4} + \frac{1}{132x^6(1+x)^5} + \frac{1}{132x^5(1+x)^6} + \frac{1}{132x^4(1+x)^7} + \frac{1}{132x^3(1+x)^8} + \frac{1}{132x^2(1+x)^9} + \frac{1}{132x(1+x)^{10}} + \frac{1}{132x^{10}(1+x)} - O\left(\frac{1}{12x^{12}(1+x)} + \frac{1}{12x^{11}(1+x)^2}\right.$$

$$+\frac{1}{12\,x^{10}\,(1+x)^3}+\frac{1}{12\,x^9\,(1+x)^4}+\frac{1}{12\,x^8\,(1+x)^5}+\frac{1}{12\,x^7\,(1+x)^6}$$
$$+\frac{1}{12\,x^6\,(1+x)^7}+\frac{1}{12\,x^5\,(1+x)^8}+\frac{1}{12\,x^4\,(1+x)^9}+\frac{1}{12\,x^3\,(1+x)^{10}}$$
$$+\frac{1}{12\,x^2\,(1+x)^{11}}+\frac{1}{12\,x\,(1+x)^{12}}\Bigg)+\int_x^\infty \frac{1}{t\,(t+1)}\,\mathrm{d}t$$

We don't have very explicit bounds, but it's often the case that successive terms of the Euler-Maclaurin series (after the first few) have opposite signs, and typically the actual tail T(x) is between the Euler-Maclaurin sums for $2\,n$ and $2\,n+2$. For example, in this one I claim that the following three values are in increasing order:

```
> L := [convert(eulermac(f1(t),t=x..infinity,8),polynom),
  sum(f1(t),t=x..infinity),
  convert(eulermac(f1(t),t=x..infinity,10),polynom)] assuming x
  > 0;
```

$$L:=\Bigg[\frac{1}{2\,x+2\,x^2}+\frac{1}{12\,x^2+12\,x^3}-\frac{1}{120\,x^4+120\,x^5}-\frac{1}{120\,x^3+240\,x^4+120\,x^5}$$
$$-\frac{1}{120\,x^2+360\,x^3+360\,x^4+120\,x^5}-\frac{1}{120\,x+480\,x^2+720\,x^3+480\,x^4+120\,x^5}$$
$$+\frac{1}{12\,x+24\,x^2+12\,x^3}+\frac{1}{252\,x^6+252\,x^7}+\frac{1}{252\,x^5+504\,x^6+252\,x^7}$$
$$+\frac{1}{252\,x^4+756\,x^5+756\,x^6+252\,x^7}$$
$$+\frac{1}{252\,x^2+1260\,x^3+2520\,x^4+2520\,x^5+1260\,x^6+252\,x^7}$$
$$+\frac{1}{252\,x^3+1008\,x^4+1512\,x^5+1008\,x^6+252\,x^7}$$
$$+\frac{1}{252\,x+1512\,x^2+3780\,x^3+5040\,x^4+3780\,x^5+1512\,x^6+252\,x^7}$$
$$-\frac{1}{240\,x^8+240\,x^9}-\frac{1}{240\,x^7+480\,x^8+240\,x^9}-\frac{1}{240\,x^6+720\,x^7+720\,x^8+240\,x^9}$$
$$-\frac{1}{240\,x^5+960\,x^6+1440\,x^7+960\,x^8+240\,x^9}$$
$$-\frac{1}{240\,x^4+1200\,x^5+2400\,x^6+2400\,x^7+1200\,x^8+240\,x^9}$$
$$-\frac{1}{240\,x^3+1440\,x^4+3600\,x^5+4800\,x^6+3600\,x^7+1440\,x^8+240\,x^9}$$
$$-\frac{1}{240\,x^2+1680\,x^3+5040\,x^4+8400\,x^5+8400\,x^6+5040\,x^7+1680\,x^8+240\,x^9}$$
$$-1\Big/\big(240\,x+1920\,x^2+6720\,x^3+13440\,x^4+16800\,x^5+13440\,x^6+6720\,x^7$$

$$+ 1920 x^8 + 240 x^9) + \ln(1+x) - \ln(x), \frac{1}{x}, \ln(1+x) - \ln(x)$$

$$- \frac{1}{240 x^2 + 1680 x^3 + 5040 x^4 + 8400 x^5 + 8400 x^6 + 5040 x^7 + 1680 x^8 + 240 x^9}$$

$$+ \frac{1}{2 x + 2 x^2} + \frac{1}{12 x^2 + 12 x^3} - \frac{1}{120 x^4 + 120 x^5} - \frac{1}{120 x^3 + 240 x^4 + 120 x^5}$$

$$- \frac{1}{120 x^2 + 360 x^3 + 360 x^4 + 120 x^5} - \frac{1}{120 x + 480 x^2 + 720 x^3 + 480 x^4 + 120 x^5}$$

$$+ \frac{1}{12 x + 24 x^2 + 12 x^3} + \frac{1}{252 x^6 + 252 x^7} + \frac{1}{252 x^5 + 504 x^6 + 252 x^7}$$

$$+ \frac{1}{252 x^4 + 756 x^5 + 756 x^6 + 252 x^7}$$

$$+ \frac{1}{252 x^2 + 1260 x^3 + 2520 x^4 + 2520 x^5 + 1260 x^6 + 252 x^7}$$

$$+ \frac{1}{252 x^3 + 1008 x^4 + 1512 x^5 + 1008 x^6 + 252 x^7}$$

$$+ \frac{1}{252 x + 1512 x^2 + 3780 x^3 + 5040 x^4 + 3780 x^5 + 1512 x^6 + 252 x^7}$$

$$- \frac{1}{240 x^8 + 240 x^9} - \frac{1}{240 x^7 + 480 x^8 + 240 x^9} - \frac{1}{240 x^6 + 720 x^7 + 720 x^8 + 240 x^9}$$

$$- \frac{1}{240 x^5 + 960 x^6 + 1440 x^7 + 960 x^8 + 240 x^9}$$

$$- \frac{1}{240 x^4 + 1200 x^5 + 2400 x^6 + 2400 x^7 + 1200 x^8 + 240 x^9}$$

$$- \frac{1}{240 x^3 + 1440 x^4 + 3600 x^5 + 4800 x^6 + 3600 x^7 + 1440 x^8 + 240 x^9} - 1 /$$

$$(240 x + 1920 x^2 + 6720 x^3 + 13440 x^4 + 16800 x^5 + 13440 x^6 + 6720 x^7 + 1920 x^8$$

$$+ 240 x^9) + \frac{1}{132 x^{10} + 132 x^{11}} + \frac{1}{132 x^9 + 264 x^{10} + 132 x^{11}}$$

$$+ \frac{1}{132 x^8 + 396 x^9 + 396 x^{10} + 132 x^{11}}$$

$$+ \frac{1}{132 x^7 + 528 x^8 + 792 x^9 + 528 x^{10} + 132 x^{11}}$$

$$+ \frac{1}{132 x^6 + 660 x^7 + 1320 x^8 + 1320 x^9 + 660 x^{10} + 132 x^{11}}$$

$$+ \frac{1}{132 x^5 + 792 x^6 + 1980 x^7 + 2640 x^8 + 1980 x^9 + 792 x^{10} + 132 x^{11}}$$

$$+ \frac{1}{132 x^4 + 924 x^5 + 2772 x^6 + 4620 x^7 + 4620 x^8 + 2772 x^9 + 924 x^{10} + 132 x^{11}}$$

$+1/(132 x^2 + 1188 x^3 + 4752 x^4 + 11088 x^5 + 16632 x^6 + 16632 x^7 + 11088 x^8$

$+ 4752 x^9 + 1188 x^{10} + 132 x^{11})$

$+1/(132 x^3 + 1056 x^4 + 3696 x^5 + 7392 x^6 + 9240 x^7 + 7392 x^8 + 3696 x^9$

$+ 1056 x^{10} + 132 x^{11}) + 1/(132 x + 1320 x^2 + 5940 x^3 + 15840 x^4 + 27720 x^5$

$+ 33264 x^6 + 27720 x^7 + 15840 x^8 + 5940 x^9 + 1320 x^{10} + 132 x^{11})]$

I'm going to need to increase Digits somewhat, because roundoff error will be ugly.

```
> Digits:= 17: plot([(L[2]-L[1])*x^11,(L[3]-L[2])*x^13],x=1.
  .10,colour=[red,blue]);
```



To get the best possible approximation for our sum (with a fixed x) using Euler-Maclaurin series, we take more and more terms until the values stop getting closer together. I'll try it for $x=2$: we want the best Euler-Maclaurin approximation for $\sum_{j=2}^{\infty} \frac{1}{j\,(j+1)}$ (the actual value is 1/2).

```
> eval([seq([2*n,eulermac(f1(t),t=x..infinity,2*n)],n=1..20)],
```

```
{x=2,O=0});
```

$$\left[\left[2, \ln(3) - \ln(2) + \frac{41}{432}\right], \left[4, \ln(3) - \ln(2) + \frac{2939}{31104}\right], \left[6, \ln(3) - \ln(2) + \frac{158801}{1679616}\right], \left[8, \ln(3) - \ln(2) + \frac{7621187}{80621568}\right], \left[10, \frac{34297979}{362797056} + \ln(3) - \ln(2)\right], \left[12, \frac{14815926059}{156728328192} + \ln(3) - \ln(2)\right], \left[14, \frac{88900322939}{940369969152} + \ln(3) - \ln(2)\right], \left[16, \frac{25601464030595}{270826551115776} + \ln(3) - \ln(2)\right], \left[18, \frac{207397397708150 5}{21936950640377856} + \ln(3) - \ln(2)\right], \left[20, -\ln(2) + \ln(3) + \frac{8293682643532787}{87747802561511424}\right], \left[22, -\ln(2) + \ln(3) + \frac{149392263628263571}{1579460446107205632}\right], \left[24, -\ln(2) + \frac{6439074967833189881 9}{68232691271831283302 4} + \ln(3)\right], \left[26, -\ln(2) + \ln(3) + \frac{4871114430207300584 3}{51174518453873462476 8}\right], \left[28, -\ln(2) + \ln(3) + \frac{1349353358295740358094 1}{1473826131471555719331 84}\right], \left[30, -\ln(2) + \ln(3) + \frac{877282238957315851728289}{7958661109946400884391936}\right], \left[32, -\ln(2) + \ln(3) + \frac{37354963940875597672217 9}{152806293310970896980325171 2}\right], \left[34, -\ln(2) + \frac{101183088866771821716533161 51}{137525663979873807282292654 08} + \ln(3)\right], \left[36, -\ln(2) + \ln(3) - \frac{1070910363076726918083860238461 5}{222791575647395567797314099609 6}\right], \left[38, -\ln(2) + \ln(3) + \frac{2076631911066977860609382780073 5}{495092390327545706216253554688}\right], \left[40, -\ln(2) + \ln(3) - \frac{2546084523445434288530818976557250 81}{641639737864499235256264606875648}\right]\right]$$

```
> L:=evalf(%);
```

$L := [[2., 0.50037251551557180], [4., 0.49995456284067468], [6., 0.50001112338784736], [8., 0.49999548241197352], [10., 0.50000275231661129], [12., 0.49999764239800739], [14., 0.50000271123809829], [16., 0.49999595792742969], [18., 0.50000759995587279], [20., 0.49998237693343093], [22., 0.50004947329214925], [24., 0.49983446168064254], [26., 0.50065143476794364], [28., 0.49701955469328944], [30., 0.51569498557598687], [32., 0.40570956769900634], [34., 1.1412047943501407], [36., -4.4013161176398831], [38., 42.349796138789312], [40., -396.40358102723478]]$

By the way, why didn't I just use **eulermac(f1(t),t=2..infinity,2*n)**? Because it wouldn't work right: **eulermac** needs the interval to involve a parameter that's supposed to go to infinity.

```
> eulermac(f1(t),t=2..infinity,2);
```

$$\ln(2) + \gamma - \frac{79}{120} + O(1)$$

```
> [seq([j,L[j,2]-L[j-1,2]],j=2..20)];
```

[[2, -0.00041795267489712], [3, 0.00005656054717268], [4, -0.00001564097587384], [5, 0.00000726990463777], [6, -0.00000510991860390], [7, 0.00000506884009090], [8, -0.00000675331066860], [9, 0.00001164202844310], [10, -0.00002522302244186], [11, 0.00006709635871832], [12, -0.00021501161150671], [13, 0.00081697308730110], [14, -0.00363188007465420], [15, 0.01867543088269743], [16, -0.10998541787698053], [17, 0.73549522665113436], [18, -5.5425209119900238], [19, 46.751112256429195], [20, -438.75337716602409]]

```
> map(t -> abs(t[2]),%);
```

[0.00041795267489712, 0.00005656054717268, 0.00001564097587384, 0.00000726990463777, 0.00000510991860390, 0.00000506884009090, 0.00000675331066860, 0.00001164202844310, 0.00002522302244186, 0.00006709635871832, 0.00021501161150671, 0.00081697308730110, 0.00363188007465420, 0.01867543088269743, 0.10998541787698053, 0.73549522665113436, 5.5425209119900238, 46.751112256429195, 438.75337716602409] **(4.3)**

```
> select(has,%%,{min(%),-min(%)});
```

[[7, 0.00000506884009090]] **(4.4)**

The smallest absolute difference is at j=7, corresponding to L[6] and L[7]

```
> L[6],L[7];
```

[12., 0.49999764239800739], [14., 0.50000271123809829]

So the true value should be somewhere between these. We might try the average of these.

```
> (L[6,2]+L[7,2])/2;
```

0.50000017681805284 **(4.5)**

Maple commands introduced in this lesson:

min
logplot (in **plots** package)
cosh
bernoulli
eulermac